
SELF?

SaL Error Logging Facility

® ajuta a capire, analizzare e risolvere i problemi con gli errori e gli
avvertimenti restituiti dalle istruzioni SQL

» SELF catturai dettagli per categorie di errore o specifici codici di
errore o avvertimenti SQL

® Pyo essere attivato per una specifica sessione SQL o per tutte |e
sessioni di tutti i job del sistema

https://www.ibm.com/docs/en/i/7.52topic=to da 7.4 TRy
ols-sql-error-logging-facility-self 7.5 TR1

Cosafa?

® Quando viene attivato
= controlla gli esiti SQLCODE delle istruzioni SQL

® se SQLCODE coincide con uno di quelli che si & scelto di monitorare viene
aggiunto o aggiornato un record nella tabella di log QSYS2 . SQL_ERRORT
con le informazioni sull'errore e il contesto (istruzione SQL, stack chiamate,
utente, ecc.)

Come e fatto?

» Variabile globale SYSIBMADM.SELFCODES: contiene la lista
degli sqlcode che si desidera controllare

® | 'impostazione di un valore di default nella creazione della variabile globale
consente di attivare il monitoraggio per qualsiasi job

® Per un monitoraggio in una specifica sessione SQL occorre eseguire
'istruzione SQL set per modificare il valore della variabile globale

https://www.ibm.com/docs/en/i/7.5?topic=services-selfcodes-global-variable

» Vista QSYS2.SQL_ERROR_LOG: perinterrogare tutto quanto e
stato intercettato tramite SELF

https://www.ibm.com/docs/en/i/7.5?topic=services-sql-error-log-view

Come si attiva?

» (Occorre settare la variabile globale SYSIBMADM.SELFCODES
con l'elenco degli SQLCODE che si desidera monitorare separati da
, (virgola)

®» Numero massimo di specifici SQLCODE monitorabili: 32

® | 3 funzione VALIDATE SELF consente di controllare se il valore
settato in SELFCODES e sintatticamente corretto

-- attivazione controllo a livello globale
create or replace variable SYSIBMADM.SELFCODES varchar(256)

default (SYSIBMADM.VALIDATE_SELF('-913, -206"));

-- attivazione controllo nella sessione SQL corrente
set SYSIBMADM.SELFCODES =
(SYSIBMADM.VALIDATE_SELF('-913, -206"'));

Gruppi di errori

®» |nvece di elencare i singoli codici di errore e possibile settare la
variabile globale SELFCODES con uno dei 3 valori speciali

» xERROR: tutti i valori di sqglcode per gli errori

» xWARN: tutti i valori di sqlcode per gli avvertimenti

= +xALL: tutti i valori di sqlcode per errori e avvertimenti

-- solo errors

set SYSIBMADM.SELFCODES = 'xERROR';
-- solo warnings

set SYSIBMADM.SELFCODES = 'xWARN';
-- entrambi

set SYSIBMADM.SELFCODES = 'xALL';
-- nessun controllo

set SYSIBMADM.SELFCODES = '«NONE'; — |

E equivalente settarea @ o null

E' attivo?

® Per verificare se la registrazione SELF € attiva occorre visualizzare
il contenuto della variabile globale SYSIBMADM.SELFCODES

-- verifica sqglcode monitorati
values SYSIBMADM.SELFCODES;

Comessileggeil log?

®» |nterrogare la vista SQL_ERROR_LOG

select *
from QSYS2.SQL_ERROR_LOG
order by LOGGED_TIME desc;

e

123 LOGGED_SQLCODE

AEC | OGGED_SQLSTATE
123 NUMBER_OCCURRENCES

STATEMENT_TEXT
asc STATEMENT_OPERATION

ABC STATEMENT_OPERATION_DETAIL

123 REASON_CODE

AoC PROGRAM_LIBRARY
Aec PROGRAM_NAME

Aoc PROGRAM_TYPE

#t MODULE_NAME

9 LOGGED_TIME

AoC JOB_NAME

123 THREAD._ID

#:c ADORTED USER NAME

-~ 2
CRE S e T

-- simulazione errorelSL

El

EXECUTE IMMEDIATE

Qsys

QZDASRV

*SRVPGM

QZDASQL

2024-07-20 10:47:39.204000
076525/QUSER/QZDASQINIT

ROM EMPLOYEE

= . T, F N S N

e N A

 aec USER_NAME

ABC SYSTEM_USER_MNAME

ABC CLIENT_ACCTNG

ABC CLIENT_APPLNAME

ABC CLIENT_PROGRAMID

RBC CLIENT_USERID

ABC CLIENT_WRKSTNNAME

REC RDB_MAME

2 INITIAL_LOGGED_TIME

AEC INITIAL_JOB_NAME

123 INITIAL_THREAD_ID

ARC INITIAL_ADOPTED_USER_NAME

] INITIAL_STACK

Per la lista dei codici
statement_operation

cfr. campo QQC21 in
Database Monitor view 1000

AP PN AN AV ALY,

2024-07-20 10:47:39.204000
076525/QUSER/QZDASOINIT
0

P T N N i L S A
- ~ Y, - N N A A R N

{"Initial_stack"[{"ORD™11,"TYPE""ILE","LIB""QSYS","PGM":"QSQXCUTE","MODULE":"QSQXCL

Ripetizione della stessa istruzione

® Quando SELF intercetta un errore puo inserire un nuovo record o
aggiornare un record gia esistente nella tabella SQL_ERRORT

® || record viene aggiornato in base a questa chiave:
™ | ibreria programma
= Nome programma
= Nome modulo
= SQLCODE
™ |struzione SQL

® |n caso di aggiornamento viene incrementato il campo
NUMBER_OCCURRENCES

= | campi con prefisso INITIAL _ siriferiscono alla prima registrazione del
log

Come si pulisce?

® Periodicamente e opportuno cancellare i record obsoleti dalla
tabella QSYS2.SQL_ERRORT

-- pulizia log piu vecchi di 30 gg
delete from QSYS2.SQL_ERRORT
where LOGGED_TIME <= date(current date - 30 days);

Come si usa?

» (Caso 1) attivazione globale
e sufficiente settare il valore di default della variabile globale
da questo momento SELF e attivo per tutti i job

® (Caso 2) attivazione per job corrente
occorre settare la variabile globale nella sessione SQL corrente
con istruzione SQL set

ATTENZIONE!

Se in una sessione interattiva 5250 si apre l'interfaccia STRSQL per eseguire l'istruzione
set per attivare SELF, appena si esce con F3 da strsql la sessione SQL termina e quindi
anche il valore settato nella variabile globale SELFCODES ritorna al suo valore di default

Attivazione dinamica SELF

® Se si desidera attivare SELF solo in un contesto specifico si puo
ricorrere a due metodi:

1. Modificare il programma in esame aggiungendo l'istruzione set
della variabile globale SELFCODES al momento in cui si necessita
di attivare SELF

2. Predisporre tutti i programmi embedded SQL ad esequire
'istruzione set della variabile globale SELFCODES condizionata
da una configurazione di attivazione dipendente dal job (p.es. con
una variabile di ambiente) o dal contesto (p.es. un'area dati
presente in lista librerie)

Attivare SELF con una variabile di ambiente/a

® ysiamo una variabile di ambiente (p.es. SELF_ACTIVATION)
settata a livello di job perinformare i programmi che andranno in
esecuzione se devono attivare o meno SELF

/* Imposto i1l valore della variabile di ambiente */
ADDENVVAR ENVVAR(SELF_ACTIVATION) VALUE('*YES')
LEVEL (¥JOB) REPLACE(*YES)

// nel programma RPG dichiaro prototipo e variabili per reperire
// 11 valore della variabile di ambiente
dcl-pr getEnv pointer extproc('QpOzGetEnvNoCCSID');
*n pointer value options(*xstring);
end-pr;
dcl-s pEnvVar pointer;
dcl-s rtnEnvVarValue varchar(10) inz(xblanks);
dcl-c EnvVarName 'SELF_ACTIVATION';
dcl-s SelfStatus varchar(256);
dcl-s Msg varchar(1000);




Attivare SELF con una variabile di ambiente/2

rtnEnvVarValue = %str(pEnvVar);

[(—
pEnvVar = getEnv(EnvVarName); /////
1f pEnvVar <> *null;

endif; imposto il tipo di
select; attivazione di SELF in
when rtnEnvVarValue = '%xYES'; base al valore della
exec sql N : :

set SYSIBMADM.SELFCODES = '*ERROR'; variabile di ambiente
when rtnEnvVarValue = '«NO' or rtnEnvVarValue = xblanks;
exec sql

set SYSIBMADM.SELFCODES = 'xNONE';
when rtnEnvVarValue <> xblanks;

reperisco valore variabile di ambiente

exec sql
set SYSIBMADM.SELFCODES = sysibmadm.validate_self(:rtnEnvVarValue);
endsl; _
exec sql
values SYSIBMADM.SELFCODES into :SelfStatus; ////\\\\
Msg = 'Stato di attivazione SELF: '+ SelfStatus; informo nel joblog sullo
snd-msg Msg; stato di attivazione di

Power coffee - MK1 Vedi esempio allegato TESTSELF . SQLRPGLE

Best practice

® Porre attenzione ad attivare SELF su larga scala

® Per esempio settando il valore di default (= tutti i job) della
variabile globale SELFCODES uguale a *ALL (=tutti i possibili
codici di errore e warning)

® Si potrebbe ottenere |'effetto di scrivere/aggiornare numerosi
record da diversi job contemporaneamente nella tabella di log
SQL_ERRORT con potenziali problemi di performance dovuti a

lock concorrenti ']

= Usare ma non abusare D

Riferiment;

|\7‘olutlon ® F-mail aziendale: marco.riva@ivolution.it

m » Blog: www.markonetools.it

® F-mail blog: info@markonetools.it

m ®» | inkedin: www.linkedin.com/in/marcoriva-mka
, ®» Twitter: @MarcoRivaz3

® Facebook: https://www.facebook.com/markonetools/

Youg[/li: ® YouTube: youtube.com/@markonetools

**free



  // ESEMPIO SQL DINAMICO CON USO DI UN PARAMETER MARKER PER OUTPUT

  //  esempio con errori per test SELF

  //  vedi specifiche tra i commenti ++SELF-START e ++SELF-END



  // esempio impostazione variabile di ambiente per attivare SELF

  // ADDENVVAR ENVVAR(SELF_ACTIVATION) VALUE('*YES') LEVEL(*JOB) REPLACE(*YES)



  // (c) 2024 MarkOneTools - www.markonetools.it



 ctl-opt copyright('www.MarkOneTools.it')

         decedit('0,')

         indent(' ')

         option(*nodebugio: *srcstmt: *showcpy)

         expropts(*resdecpos)

         extbinint(*yes)

         dftactgrp(*no);



 dcl-s sqlstm varchar(500);

 dcl-s NRek uns(10) inz(*zeros);



 //++SELF-START

 // API reperimento valore di una variabile di ambiente a livello job

 dcl-pr getEnv pointer extproc('Qp0zGetEnvNoCCSID');

   *n pointer value options(*string);

 end-pr;

 dcl-s pEnvVar pointer;

 dcl-s rtnEnvVarValue varchar(10) inz(*blanks);

 dcl-c EnvVarName 'SELF_ACTIVATION';

 dcl-s SelfStatus varchar(256);

 dcl-s Msg varchar(1000);

 //++SELF-END



 // metodo 1 con nome tabella errato

 reset NRek;

 sqlstm = 'values(select count(*) from EMPLOYEEEEEEE) into ?';



 exec sql

   prepare STM1 from :sqlstm;



 if sqlcode < *zeros;

   dsply ('Errore preparazione: ' + %char(sqlcode));

 else;

   exec sql

     execute STM1 using :NRek;



   dsply ('Numero impiegati: ' + %char(NRek));

 endif;



 // metodo 2 --> con sintassi errata per embedded SQL

 reset NRek;

 sqlstm = 'select count(*) into ? from EMPLOYEE';



 exec sql

   prepare STM1 from :sqlstm;



 if sqlcode < *zeros;

   dsply ('Errore preparazione: ' + %char(sqlcode));

 else;

   exec sql

     execute STM1 using :NRek;



   dsply ('Numero impiegati: ' + %char(NRek));

 endif;



 *inlr = *on;

 return;



 // subroutine iniziale

 begsr *inzsr;



   //++SELF-START

   // reperisco informazione da variabile ambiente per

   //  sapere se attivare SELF

   pEnvVar = getEnv(EnvVarName);

   if pEnvVar <> *null;

     rtnEnvVarValue = %str(pEnvVar);

   endif;

   // tipo di attivazione SELF

   select;

     when rtnEnvVarValue = '*YES';

       exec sql

         set SYSIBMADM.SELFCODES = '*ERROR';

     when rtnEnvVarValue = '*NO' or rtnEnvVarValue = *blanks;

       exec sql

         set SYSIBMADM.SELFCODES = '*NONE';

     when rtnEnvVarValue <> *blanks;

       exec sql

         set SYSIBMADM.SELFCODES = sysibmadm.validate_self(:rtnEnvVarValue);

   endsl;

   exec sql

     values SYSIBMADM.SELFCODES into :SelfStatus;

   Msg = 'Stato di attivazione SELF: '+ SelfStatus;

   snd-msg Msg;

   //++SELF-END



 endsr;

